GAMS

Transportation Model

Jan-Hendrik Jagla jhjagla@gams.com

Lutz Westermann lwestermann@gams.com

GAMS Development Corporation

www.gams.com

GAMS Software GmbH

www.gams.de

What is a Model?

- Mathematical Programming (MP) Model
  - List of Equations

- Collection of several intertwined MP Models
  - Data Preparation
  - Data Calibration
  - “Solution” Module (e.g. sequential, parallel, loop)
  - Report Module

A Transportation Model



Minimize	Transportation cost
subject to	Demand satisfaction at markets
	Supply constraints

Mathematical Algebra

$$
\begin{array}{rcll}
\sum\limits_{\substack{c,p: \\ (c,p)\in\mathcal{N}}} tcost \cdot dist(c,p) \cdot x_p^c & \rightarrow & \min & \\
\sum\limits_{\substack{c,p: \\ (c,p)\in\mathcal{N}}} x_p^c & \leq & sup(c) & \forall c \\
\sum\limits_{\substack{c,p: \\ (c,p)\in\mathcal{N}}} x_p^c & \geq & dem(p) & \forall p \\
x_p^c & \geq & 0 & \forall c,p : (c,p) \in \mathcal{N}
\end{array}
$$

GAMS Algebra




```
Variables
     x(i,j)  shipment quantities in cases
     z       total transportation costs in thousands of dollars ;

Positive Variable x ;

Equations
     cost         define objective function
     supply(i)    observe supply limit at plant i
     demand(j)    satisfy demand at market j ;

cost ..         z  =e=  sum((i,j), c(i,j)*x(i,j)) ;

supply(i) ..   sum(j, x(i,j))  =l=  a(i) ;

demand(j) ..   sum(i, x(i,j))  =g=  b(j) ;

Model transport /all/ ;
```

A few Word about GAMS Syntax

- Symbols:
  - Sets
  - Parameters
  - Variables
  - Equations
  - Models
  - ASCII Output Files

```
Sets        i           canning plants   / seattle, san-diego /;
Parameters  a(i)        capacity of plant i in cases
         /  seattle      350
            san-diego    600   /;
Variables   x(i,j)      shipment quantities in cases;
Equations   supply(i)   observe supply limit at plant i;
Model       transport   /all/ ;
File        fx          some file / 'c:\t\text.txt' /
```

- Statements
  - Declarations
  - Data Assignments
  - Equation Definition
  - Programming Flow Control
  - Option statement

```
Parameter c(i,j);
c(i,j) = f * d(i,j) / 1000 ;
supply(i) ..   sum(j, x(i,j))  =l=  a(i);
loop(i, put fx i.tl);
option reslim=10;
```

Using the GAMS IDE to build a model

- IDE Project Management
- Documentation
  - User's Guide
  - McCarl User's Guide
  - Solver Manuals
- Model Library
- Solver Selection
- Option Editor
- Listing file
  - Tree view
  - Error navigation
- Spell checking



Hands-on! Transportation Model



Types of Variables

- Continuous Variables
  - Free/Positive/Negative
  - Lower and/or upper bound
- Binary Variables
  - Either 0 or 1
- Integer Variables
  - by default ranging between 0 and 100
- Semicont/Semiint Variables
  - 0 or above a given minimum
- Special Ordered Set Variables (SOS1, SOS2)

Binary Variables

- Powerful Tool to model yes/no decisions
- Models with discrete variables (MIP)
  - Solved using Branch-and-Cut algorithms (lots or LPs)
  - Theoretically difficult problem class
  - Practical:
    - mixed bag
    - *Art of Modeling*
- Example: Minimum Shipment
  - Ship at least 100 tons or don't ship

Minimum Shipment

- Continuous Variable *x* (shipment)

- Binary Variable *ship* (indicator for minimum shipment):
  - ship=1 if $x \geq 100$
  - ship=0 if $x = 0$

- Coupling Constraints:
  - $x \geq 100 * ship$
  - $x \leq \text{bigM} * ship$

- How big do we have to make bigM?

Implement Min/Max Shipments (MIP)

```
Parameter rep1(i,j,*)    Shipments between plants and markets
          rep2(*)        Objective value;

rep1(i,j,'lp') = x.l(i,j);
rep2('lp')     = z.l;

scalars xmin / 100 /
        xmax / 275 /;

binary variables ship(i,j)    decision variable to ship
equations        minship(i,j) minimum shipments
                 maxship(i,j) maximum shipments ;

minship(i,j).. x(i,j) =g=   xmin*ship(i,j);
maxship(i,j).. x(i,j) =l=   xmax*ship(i,j);

model m2 min shipments / all /;
solve m2 using mip minimizing z;
rep1(i,j,'mip') = x.l(i,j);
rep2('mip')     = z.l;

option mip=coincbc
solve m2 using mip minimizing z;
rep1(i,j,'mip-coincbc') = x.l(i,j);
rep2('mip-coincbc')     = z.l;
display rep1,rep2;
```

Economy of Scales: $Cost = Volume^{factor}$



Implement Nonlinear Cost (NLP)

```
* nonlinear cost
equation nlcost nonlinear cost function;
scalar beta;

nlcost.. z =e= sum((i,j), c(i,j)*x(i,j)**beta);

model m3 / nlcost,supply,demand /;

beta = 1.5;
solve m3 using nlp minimizing z;
rep1(i,j,'nlp-convex') = x.l(i,j);
rep2('nlp-convex')     = z.l;

beta = 0.6;
solve m3 using nlp minimizing z;
rep1(i,j,'nlp-concave') = x.l(i,j);
rep2('nlp-concave')     = z.l;

option nlp=baron;
solve m3 using nlp minimizing z;
rep1(i,j,'nlp-baron') = x.l(i,j);
rep2('nlp-baron')      = z.l;

display rep1,rep2;
```

Implement Min/Max and Nonlinear (MINLP)

```
* min/max and nonlinear objective

model m4 / nlcost,supply,demand,minship,maxship /;

option minlp=baron;
solve m4 using minlp minimizing z;
rep1(i,j,'minlp-bar') = x.l(i,j);
rep2('minlp-bar')     = z.l;

option minlp=lindoglobal;
solve m4 using minlp minimizing z;
rep1(i,j,'minlp-lin') = x.l(i,j);
rep2('minlp-lin')     = z.l;

display rep1,rep2;
```

Sources of GAMS Information

Download: http://download.gams-software.com/
Release Notes: http://www.gams.com/docs/release/release.htm
Contributed Documentation: http://www.gams.com/docs/contributed
Contributed Software: http://www.gams.com/contrib/contrib.htm
Presentations: http://www.gams.com/presentations
Workshops: http://www.gams.com/courses.htm

Bruce McCarl's Newsletter: http://www.gams.com/maillist/newsletter.htm
GAMS User Group: http://www.gams.com/maillist/gams_l.htm
GAMS Google Group: http://groups.google.de/group/gamsworld

Other relevant sites on the Web: http://www.gams.com/hotlinks.htm

Contacting GAMS

USA

GAMS Development Corp.
1217 Potomac Street, NW
Washington, DC 20007
USA
Phone: +1 202 342 0180
Fax: +1 202 342 0181
http://www.gams.com

sales@gams.com
support@gams.com

Europe

GAMS Software GmbH
Eupener Str. 135-137
50933 Cologne
Germany
Phone: +49 221 949 9170
Fax: +49 221 949 9171
http://www.gams.de

info@gams.de
support@gams-software.com